
A Taylor Function Calculus for Hybrid System
Analysis

Validation in Coq

P. Collins1 M. Niqui1 N. Revol2

1CWI

2INRIA, LIP, Université de Lyon

Motivation

Verification of Hybrid Systems.

• Ariadne: Tool for analysis of nonlinear hybrid systems.
http://trac.parades.rm.cnr.it/ariadne/

• Example: Performing reachability analysis.

Motivation

Main steps:

Computing the flow of a differential equations ẋ = f (x)

Computing an outer-approximation of an enclosure on a grid.

These operations must be performed rigorously and efficiently.

Ariadne

implemented in (C++)
kernel is generic!, there are theories for floats, reals and
continuous functions.
theories still work without full implementation.

To get validated results:

verify the kernel
I primitives for function calculus
I algorithms for reachability analysis

verify the result of each calculation.

We use the Coq system.

Validating Ariadne

Numerals

Functions

Reachability
Algorithms

Validated
Hyb. Sys.

Floats
p-adic
Rationals
Reals

Polynomials
Taylor Model
Power Series

Taylor Models (TM)

Approximate functions using their Taylor expansion.

f : [−1,1] −→ R approximated by Tf : [−1,1] −→ R

f(x)
T(x)
f

Taylor Models (TM)

Approximate functions using their Taylor expansion.

f : [−1,1] −→ R approximated by Tf : [−1,1] −→ R

Each polynomial T approximates a family of functions.

T(x)

f(x)

Verifying TM

1 TM using constructive reals in Coq Zumkeller
2 TM using rational intervals in PVS Cháves–Daumas
3 Chebyshev models in HOL+Coq ... Joldes, Mayero

We implement a basic calculus of Taylor models with coefficients from
an abstract data-type F.

Verifying TM

1 TM using constructive reals in Coq Zumkeller
2 TM using rational intervals in PVS Cháves–Daumas
3 Chebyshev models in HOL+Coq ... Joldes, Mayero

We implement a basic calculus of Taylor models with coefficients from
an abstract data-type F.

F: the minimum interface with respect to which we have a basic Taylor
model calculus.

It covers Floats (various base/precision), arbitrary precision, exact etc.

Taylor + Floats in Coq NSV-3 8 / 25



Numerals

F : Type

Constant:
0F : F

Operations:

− : F −→ F opposite
| | : F −→ F abs. value
⊕u,⊕d ,⊕n : F→ F −→ F rounded addition
⊗u,⊗d ,⊗n : F→ F −→ F rounded multiplication

Injection

: F −→ R

Numerals

Axioms:

• 0F = 0
• ∀z, −z = −z

• ∀z, |z| = |z|

• ∀z0z1, |z0 ⊕n z1 − (z0 + z1)| ≤ |z0 ⊕u z1 − (z0 + z1)|
• ∀z0z1, |z0 ⊕n z1 − (z0 + z1)| ≤ |(z0 + z1)− z0 ⊕d z1|
• ∀z0z1, z0 ⊕d z1 ≤ z0 + z1 ≤ z0 ⊕u z1

• ∀z0z1, |z0 ⊗n z1 − (z0 · z1)| ≤ |z0 ⊗u z1 − (z0 · z1)|
• ∀z0z1, |z0 ⊗n z1 − (z0 · z1)| ≤ |(z0 · z1)− z0 ⊗d z1|
• ∀z0z1, z0 ⊗d z1 ≤ z0 · z1 ≤ z0 ⊗u z1

Axiomatisation

• Formalised in Coq
• Instantiations not needed in our work
• Possible instances

Coq’s (unbound) Floats Daumas–Rideau–Théry, Boldo
Subsets of F32, F64 (normalised, no NaN,±∞ etc.)
p-adics
Q
R (axiomatic, non-constructive)
Constructive exact reals
Singleton {0} (in fact finite groups!)

Axiomatisation

+ Axioms for ⊕

• ∀z0z1, |z0 ⊕n z1 − (z0 + z1)| ≤ |z0 ⊕u z1 − (z0 + z1)|
• ∀z0z1, |z0 ⊕n z1 − (z0 + z1)| ≤ |(z0 + z1)− z0 ⊕d z1|

can be replaced by either of

1 ∀z0z1, |z0 ⊕n z1 − (z0 + z1)| ≤ (z0⊕uz1)	(z0⊕d z1)
2

2 ∀z0z1z, |z0 ⊕n z1 − (z0 + z1)| ≤ |z − (z0 + z1)|

3 alternatively we could define

z0 ⊕u z1 := inf{z ∈ F|z0 + z1 ≤ z}

and require that the infimum exists.

Axiomatisation

+ We can add similar axiom schema for composite operations. (eg.
fusedMultiplyAdd xy+z):

For ∗ : R× · · · × R −→ R we can add

~u,d ,n : F× · · · × F −→ F

satisfying

~d (z0, . . . , zk ) ≤ ∗(z0, . . . , zk ) ≤ ~u(z0, . . . , zk )

|~n (z0, . . . , zk )−∗(z0, . . . , zk )| ≤ |~u,d (z0, . . . , zk )−∗(z0, . . . , zk )|

Verifying TM

1 TM using constructive reals in Coq Zumkeller
2 TM using rational intervals in PVS Cháves–Daumas
3 Chebyshev models in HOL+Coq ... Joldes, Mayero

We implement a basic calculus of Taylor models with coefficients from
an abstract data-type F.

Taylor Models with Floating point Coefficients

Analysed by Revol–Makino–Berz for COSY system.

Tedious because of several layers of rounding and truncation.
TM: pair of polynomial T and ε error.

T(x)

f(x)

T(x)+e

T(x)-e

Taylor Models with Floating point Coefficients

TM: pair of polynomial T and ε the error.

For exact Taylor models ε denotes truncation error.

1 If T̂ is obtained from T with floating point rep. for coefficients,
there is rounding error |T (x)− T̂ (x)|.

2 If z ∈ F, then T̂ (z̄) can be calculated
I exactly, or
I using operations on F +extra rounding error.

Ideally, If T models f : R −→ R we should have

TM over F in Coq

Let p be a sparse polynomial over F, eg.

p(x) := a0xn0 + a1xn1 + · · ·+ akxnk

where ni < ni+1 and ai ∈ F.

ε ∈ F

〈p, ε〉 is a Taylor model.

TMF is the type of Taylor models over F.

TM over F in Coq

〈p, ε〉 |=r f if:

∀z ∈ [−1,1], |p(z̄)− f (z̄)| ≤ ε .

Let E : TMF × F −→ F, E is an evaluation if

∀〈p, ε〉∀z̄ ∈ [−1,1], |E(〈p, ε〉, z̄)− f (z̄)| ≤ ε .

Currently Ariadne has a concrete evaluation (eval).

〈p, ε〉 |= f models f if for each z with z̄ ∈ [−1,1]

TMF calculus

Scalar multiplication

If 〈p, ε〉 |=r f , c ∈ F then

〈c ⊗n p, ε′〉 |=r c̄f

where

ε′ := |c| ⊗u ε⊕u

k⊕
i=0

c ⊗u ai 	u c ⊗d ai

and p(x) = a0xn0 + a1xn1 + · · ·+ akxnk

TMF calculus

Addition

If 〈p0, ε0〉 |=r f0, 〈p1, ε1〉 |=r f1 then

〈p0 ⊕n p1, ε
′〉 |=r f0 + f1

where

ε′ := ε0 ⊕u ε1 ⊕u
⊕

ni=mj

ai ⊗u bj 	 ai ⊗d bj

and p0(x) = a0xn0 + a1xn1 + · · ·+ akxnk ,
p1(x) = b0xm0 + b1xm1 + · · ·+ bkxml

TMF calculus

We can find ε for the following operations:

monomial product; if 〈p, ε〉 |=r f then 〈xp, ε〉 |=r xf (x)

multiplication; if 〈p0, ε0〉 |=r f0, 〈p1, ε1〉 |=r f1 then

〈p0 ⊗n p1, ε
′〉 |=r f0f1

TMF calculus

Suppose 〈p, ε〉 |= f , c ∈ F, then we can find ε′ s.t.

〈c ⊗n p, ε′〉 |= c̄f

But we need to amend the axiomatisation.

Add constants 1F, εm ∈ F, let 2F:=1F ⊕u 1F, and add axioms

• 1F = 1
• 0 < εm

• |z0 ⊗n z1 − z0 × z1| ≤ |z0 ⊗n z1| ⊗n 2F ⊗n εm

Intended meaning: εm is some value > 2ulp.

+ For addition we need axioms for ⊕.

Formal Proofs

Set Implicit Arguments.
Section Taylor_Models_Sum.

Variable F: Float.

Function pre_merge (kappa1:crr F -> crr F) (kappa2:crr F -> crr F-> crr F) (p1p2: list (nat*(crr F))* list (nat*(crr F))) 
   {measure (fun l1l2=> (length (fst l1l2) + length (snd l1l2)))%nat p1p2} : list (nat*(crr F)) :=
         match p1p2 with
         | (nil, nil) => nil
         | (nil, fn2 :: p2') => fn2 :: p2'
         | (fn1 :: p1', nil) => fn1 :: p1'
         | (fn1 :: p1' , fn2 :: p2') =>
                 match lt_eq_lt_dec (fst fn1) (fst fn2) with
                 | inleft (left _) => (fst fn1 , kappa1 (snd fn1)) :: pre_merge kappa1 kappa2 (p1',(fn2 :: p2'))
                 | inleft (right _) => (fst fn1, kappa2 (snd fn1) (snd fn2)) :: pre_merge kappa1 kappa2 (p1',p2')
                 | inright _ => (fst fn2 , kappa1 (snd fn2)) :: pre_merge kappa1 kappa2 ((fn1 :: p1'),p2')
                 end
         end.
Proof.
 intros; simpl; omega.
 intros; simpl; omega.
 intros; simpl; omega.
Time Qed.

Lemma pre_merge_eq_nil_nil: forall kappa1 kappa2, pre_merge kappa1 kappa2 (nil,nil) = nil.
Proof.
 intros kappa1 kappa2; rewrite pre_merge_equation; trivial.
Qed.

Lemma pre_merge_eq_nil_cons: forall kappa1 kappa2 fn2 p2, pre_merge kappa1 kappa2 (nil,fn2 :: p2) = fn2 :: p2.
Proof.
 intros kappa1 kappa2 fn2 p2; rewrite pre_merge_equation; trivial.
Qed.

Lemma pre_merge_eq_cons_nil: forall kappa1 kappa2 fn1 p1, pre_merge kappa1 kappa2 (fn1 :: p1, nil) = fn1 :: p1.
Proof.
 intros kappa1 kappa2 fn1 p1; rewrite pre_merge_equation; trivial.
Qed.

Lemma pre_merge_eq_cons_cons: forall kappa1 kappa2 fn1 p1 fn2 p2, 
 pre_merge kappa1 kappa2 (fn1 :: p1, fn2 :: p2) = 
                 match lt_eq_lt_dec (fst fn1) (fst fn2) with
                 | inleft (left _) => (fst fn1 , kappa1 (snd fn1)) :: pre_merge kappa1 kappa2 (p1,(fn2 :: p2))
                 | inleft (right _) => (fst fn1, kappa2 (snd fn1) (snd fn2)) :: pre_merge kappa1 kappa2 (p1,p2)
                 | inright _ => (fst fn2 , kappa1 (snd fn2)) :: pre_merge kappa1 kappa2 ((fn1 :: p1),p2)
                 end.
Proof.
 intros kappa1 kappa2 fn1 p1 fn2 p2; rewrite pre_merge_equation; trivial.
Qed.

Hint Resolve pre_merge_eq_nil_nil pre_merge_eq_nil_cons pre_merge_eq_cons_nil pre_merge_eq_cons_cons.

Lemma pre_merge_eq_1: forall kappa1 kappa2 p, pre_merge kappa1 kappa2 (p,nil) = p.
Proof.
 intros kappa1 kappa2 [|fn p']; auto.
Qed. 

Lemma pre_merge_eq_2: forall kappa1 kappa2 p, pre_merge kappa1 kappa2 (nil,p) = p. 
Proof.
 intros kappa1 kappa2 [|fn p']; auto.
Qed. 

Ltac local_tactic_prove_pre_merge_equation := 
 intros kappa1 kappa2 fn1 p1 fn2 p2 H; rewrite pre_merge_eq_cons_cons;
 destruct (lt_eq_lt_dec (fst fn1) (fst fn2)) as [[H_ | H_f] | H_f]; trivial; contradict H; omega.

Lemma pre_merge_eq_3: forall kappa1 kappa2 fn1 p1 fn2 p2, (fst fn1 < fst fn2) %nat -> 
 pre_merge kappa1 kappa2 (fn1 :: p1, fn2 :: p2) = (fst fn1 , kappa1 (snd fn1)) :: pre_merge kappa1 kappa2 (p1,(fn2 :: p2)).
Proof.
 local_tactic_prove_pre_merge_equation.
Qed. 

Lemma pre_merge_eq_4: forall kappa1 kappa2 fn1 p1 fn2 p2, (fst fn1 = fst fn2) %nat -> 
 pre_merge kappa1 kappa2 (fn1 :: p1, fn2 :: p2) = (fst fn1, kappa2 (snd fn1) (snd fn2)) :: pre_merge kappa1 kappa2 (p1,p2).
Proof.
 local_tactic_prove_pre_merge_equation.
Qed. 

Lemma pre_merge_eq_5: forall kappa1 kappa2 fn1 p1 fn2 p2, (fst fn2 < fst fn1) %nat -> 
 pre_merge kappa1 kappa2 (fn1 :: p1, fn2 :: p2) = (fst fn2 , kappa1 (snd fn2)) :: pre_merge kappa1 kappa2 ((fn1 :: p1),p2).
Proof.
 local_tactic_prove_pre_merge_equation.
Qed.

Lemma pre_merge_sorted: forall kappa1 kappa2 (p1 p2: list (nat*(crr F))), is_sorted p1 -> is_sorted p2 -> is_sorted (pre_merge kappa1 kappa2 (p1,p2)).
Proof.
 intros kappa1 kappa2.
 induction p1 as [|a1 p].
  (* nil , _ *)
  intros p2 H_ap1 H_ap2; rewrite pre_merge_eq_2; assumption.
  (* a1 :: p , _ *)
  induction p2 as [|b1 q]; intros H_ap H_bq.
   (* a1 :: p , nil *)
   rewrite pre_merge_eq_1; assumption.
   (* a1 :: p , b1 : q *)
   assert (H_p:is_sorted p); [apply is_sorted_cons_inv with (fst a1) (snd a1); rewrite <- (surjective_pairing); exact H_ap|].
   assert (H_q:is_sorted q); [apply is_sorted_cons_inv with (fst b1) (snd b1); rewrite <- (surjective_pairing); exact H_bq|].
   rewrite pre_merge_eq_cons_cons.
   destruct (lt_eq_lt_dec (fst a1) (fst b1)) as [[Hlt | Heq] | Hlt].
    (* 1 *)
    assert (hyp:is_sorted (pre_merge kappa1 kappa2 (p, b1 :: q))); [apply IHp; assumption|].
    destruct p as  [|a2 p'].
     (* a1 :: nil , b1 :: q *)
     rewrite pre_merge_eq_2; apply (is_sorted_cons (fst a1) (kappa1 (snd a1)) (b1 :: q) b1); trivial.
     (* a1 :: a2 :: p1' , b1 :: q *)
     assert (H_a12: (fst a1<fst a2)%nat); [inversion H_ap; injection H1; intros H_tmp; subst a2; assumption|].
     revert hyp.
     rewrite pre_merge_eq_cons_cons.
     destruct (lt_eq_lt_dec (fst a2) (fst b1)) as [[Hlt' | Heq'] | Hlt']; intros hyp.
      apply (is_sorted_cons (fst a1) (kappa1 (snd a1))
                  ((fst a2, kappa1 (snd a2)) :: pre_merge kappa1 kappa2 (p', b1 :: q)) 
                  (fst a2, kappa1 (snd a2)) ); simpl; trivial...
      apply (is_sorted_cons (fst a1) (kappa1 (snd a1))       
                      ((fst a2, kappa2 (snd a2) (snd b1)):: pre_merge kappa1 kappa2 (p', q))
                      (fst a2, kappa2 (snd a2) (snd b1))); simpl; trivial.
      apply (is_sorted_cons (fst a1) (kappa1 (snd a1))       
                      ((fst b1, kappa1 (snd b1)) :: pre_merge kappa1 kappa2 (a2 :: p', q))
                      (fst b1, kappa1 (snd b1))); simpl; trivial.
    (* 2 *)
    assert (hyp:is_sorted (pre_merge kappa1 kappa2 (p, q))); [apply IHp; assumption|].
    destruct p as  [|a2 p']; destruct q as [|b2 q'].
     (* a1 :: nil , b1 :: nil *)
     rewrite pre_merge_eq_nil_nil; apply is_sorted_one.

     (* a1 :: nil , b1 :: b2 :: q' *)
     assert (H_b12: (fst b1<fst b2)%nat); [inversion H_bq; injection H1; intros H_tmp; subst b2; assumption|].
     assert (H_a1_b2: (fst a1<fst b2)%nat); [rewrite Heq; assumption|].
     rewrite pre_merge_eq_2; apply (is_sorted_cons (fst a1) (kappa2 (snd a1) (snd b1)) (b2 :: q') b2); trivial.
     (* a1 :: a2 :: p' , b1 :: nil *)
     assert (H_a12: (fst a1<fst a2)%nat); [inversion H_ap; injection H1; intros H_tmp; subst a2; assumption|].
     rewrite pre_merge_eq_1; apply (is_sorted_cons (fst a1) (kappa2 (snd a1) (snd b1)) (a2 :: p') a2); trivial.
     (* a1 :: a2 :: p' , b1 :: b2 :: q' *)
     assert (H_a12: (fst a1<fst a2)%nat); [inversion H_ap; injection H1; intros H_tmp; subst a2; assumption|].
     assert (H_b12: (fst b1<fst b2)%nat); [inversion H_bq; injection H1; intros H_tmp; subst b2; assumption|].
     assert (H_a1_b2: (fst a1<fst b2)%nat); [rewrite Heq; assumption|].
     revert hyp.
     rewrite pre_merge_eq_cons_cons.
     destruct (lt_eq_lt_dec (fst a2) (fst b2)) as [[Hlt' | Heq'] | Hlt']; intros hyp.
      apply (is_sorted_cons (fst a1) (kappa2 (snd a1) (snd b1))
                  ((fst a2, kappa1 (snd a2)) :: pre_merge kappa1 kappa2 (p', b2 :: q')) 
                  (fst a2, kappa1 (snd a2)) ); simpl; trivial...
      apply (is_sorted_cons (fst a1) (kappa2 (snd a1) (snd b1))       
                      ((fst a2, kappa2 (snd a2) (snd b2)):: pre_merge kappa1 kappa2 (p', q'))
                      (fst a2, kappa2 (snd a2) (snd b2))); simpl; trivial.
      apply (is_sorted_cons (fst a1) (kappa2 (snd a1) (snd b1))
                      ((fst b2, kappa1 (snd b2)) :: pre_merge kappa1 kappa2 (a2 :: p', q'))
                      (fst b2, kappa1 (snd b2))); simpl; trivial.
    (* 3 *)
    assert (hyp:is_sorted (pre_merge kappa1 kappa2 (a1 :: p, q))); [apply IHq; assumption|].
    destruct q as  [|b2 q'].
     (* a1 :: p , b1 :: nil *)
     rewrite pre_merge_eq_1; apply (is_sorted_cons (fst b1) (kappa1 (snd b1)) (a1 :: p) a1); trivial.
     (* a1 :: p , b1 :: b2 :: q' *)
     assert (H_b12: (fst b1<fst b2)%nat); [inversion H_bq; injection H1; intros H_tmp; subst b2; assumption|].
     revert hyp.
     rewrite pre_merge_eq_cons_cons.
     destruct (lt_eq_lt_dec (fst a1) (fst b2)) as [[Hlt' | Heq'] | Hlt']; intros hyp.
      apply (is_sorted_cons (fst b1) (kappa1 (snd b1))
                  ((fst a1, kappa1 (snd a1)) :: pre_merge kappa1 kappa2 (p, b2 :: q')) 
                  (fst a1, kappa1 (snd a1)) ); simpl; trivial...
      apply (is_sorted_cons (fst b1) (kappa1 (snd b1))       
                      ((fst a1, kappa2 (snd a1) (snd b2)):: pre_merge kappa1 kappa2 (p, q'))
                      (fst a1, kappa2 (snd a1) (snd b2))); simpl; trivial.
      apply (is_sorted_cons (fst b1) (kappa1 (snd b1))       
                      ((fst b2, kappa1 (snd b2)) :: pre_merge kappa1 kappa2 (a1 :: p, q'))
                      (fst b2, kappa1 (snd b2))); simpl; trivial.
Qed.

Function pre_merge_error (p1p2: list (nat*(crr F))* list (nat*(crr F))) 
   {measure (fun l1l2=> (length (fst l1l2) + length (snd l1l2)))%nat} : list (nat*(crr F)) :=
         match p1p2 with
         | (nil, nil) => nil
         | (nil, fn2 :: p2') => nil
         | (fn1 :: p1', nil) => nil
         | (fn1 :: p1' , fn2 :: p2') =>
                 match lt_eq_lt_dec (fst fn1) (fst fn2) with
                 | inleft (left _) => pre_merge_error (p1',(fn2 :: p2'))
                 | inleft (right _) => (fst fn1, minus_up (add_up (snd fn1) (snd fn2)) (add_down (snd fn1) (snd fn2))) :: pre_merge_error (p1',p2')
                 | inright _ => pre_merge_error ((fn1 :: p1'),p2')
                 end
         end.
Proof.
 intros; simpl; omega.
 intros; simpl; omega.
 intros; simpl; omega.
Time Qed.

Lemma pre_merge_error_eq_nil_nil: pre_merge_error  (nil,nil) = nil.
Proof.
 intros ; rewrite pre_merge_error_equation; trivial.
Qed.

Lemma pre_merge_error_eq_nil_cons: forall fn p, pre_merge_error  (nil,fn :: p) = nil.
Proof.
 intros fn p; rewrite pre_merge_error_equation; trivial.
Qed.

Lemma pre_merge_error_eq_cons_nil: forall fn p, pre_merge_error  (fn :: p, nil) = nil.
Proof.
 intros fn p; rewrite pre_merge_error_equation; trivial.
Qed.

Lemma pre_merge_error_eq_cons_cons: forall fn1 p1 fn2 p2, 
 pre_merge_error  (fn1 :: p1, fn2 :: p2) = 
                 match lt_eq_lt_dec (fst fn1) (fst fn2) with
                 | inleft (left _) => pre_merge_error  (p1,(fn2 :: p2))
                 | inleft (right _) => (fst fn1, minus_up (add_up (snd fn1) (snd fn2)) (add_down (snd fn1) (snd fn2)) ) :: pre_merge_error  (p1,p2)
                 | inright _ => pre_merge_error  ((fn1 :: p1),p2)
                 end.
Proof.
 intros  fn1 p1 fn2 p2; rewrite pre_merge_error_equation; trivial.
Qed.

Hint Resolve pre_merge_error_eq_nil_nil pre_merge_error_eq_nil_cons pre_merge_error_eq_cons_nil pre_merge_error_eq_cons_cons.

Lemma pre_merge_error_eq_1: forall  p, pre_merge_error  (p,nil) = nil.
Proof.
 intros  [|fn p']; auto.
Qed. 

Lemma pre_merge_error_eq_2: forall  p, pre_merge_error  (nil,p) = nil. 
Proof.
 intros  [|fn p']; auto.
Qed. 

Ltac local_tactic_prove_pre_merge_error_equation := 
 intros  fn1 p1 fn2 p2 H; rewrite pre_merge_error_eq_cons_cons;
 destruct (lt_eq_lt_dec (fst fn1) (fst fn2)) as [[H_ | H_f] | H_f]; trivial; contradict H; omega.

Lemma pre_merge_error_eq_3: forall  fn1 p1 fn2 p2, (fst fn1 < fst fn2) %nat -> 
 pre_merge_error  (fn1 :: p1, fn2 :: p2) = pre_merge_error  (p1,(fn2 :: p2)).
Proof.
 local_tactic_prove_pre_merge_error_equation.
Qed. 

Lemma pre_merge_error_eq_4: forall  fn1 p1 fn2 p2, (fst fn1 = fst fn2) %nat -> 
 pre_merge_error  (fn1 :: p1, fn2 :: p2) = (fst fn1, minus_up (add_up (snd fn1) (snd fn2)) (add_down (snd fn1) (snd fn2))) :: pre_merge_error  (p1,p2).
Proof.
 local_tactic_prove_pre_merge_error_equation.
Qed. 

Lemma pre_merge_error_eq_5: forall  fn1 p1 fn2 p2, (fst fn2 < fst fn1) %nat -> 
 pre_merge_error  (fn1 :: p1, fn2 :: p2) = pre_merge_error  ((fn1 :: p1),p2).
Proof.
 local_tactic_prove_pre_merge_error_equation.
Qed.

Definition pre_merge_add_error (p1 p2: list (nat*(crr F))) : list (nat*(crr F)) := 

  pre_merge_error (p1,p2).

Definition error_add (t1 t2: (Taylor_model F)) : F.(crr) := add_up t1.(error) (add_up t2.(error) (sum_add_up (pre_merge_add_error t1.(spolynom) t2.(spolynom)))).

Definition pre_merge_add_near (p1 p2: list (nat*(crr F))) : list (nat*(crr F)) := pre_merge Datatypes.id (@add_near F) (p1,p2).

Definition pre_merge_add_near_sorted: forall (p1 p2: list (nat*(crr F))), is_sorted p1 -> is_sorted p2 -> is_sorted (pre_merge_add_near p1 p2) 
  := pre_merge_sorted Datatypes.id (@add_near F).

Definition merge_add_near (p1_ p2_: Sparse_polynom F) : Sparse_polynom F :=
       match p1_, p2_ with
       | Build_Sparse_polynom p1 H1, Build_Sparse_polynom p2 H2 =>
               Build_Sparse_polynom (pre_merge_add_near p1 p2) (@pre_merge_add_near_sorted p1 p2 H1 H2)
       end.

Definition add_Taylor (t1 t2: Taylor_model F) : Taylor_model F:= 
 Build_Taylor_model (merge_add_near t1.(spolynom) t2.(spolynom)) (error_add t1 t2).

Lemma sum_add_up_nonneg:forall (t1 t2: Taylor_model F), 0<= inj_R( sum_add_up (pre_merge_add_error (spolynom t1) (spolynom t2))).
Proof.
 intros [[p1 H1] e1].
 induction p1; intros [[p2 H2] e2].
  (* nil, _ *)
  simpl; unfold pre_merge_add_error; rewrite pre_merge_error_eq_2; simpl; rewrite f0_R0; auto with real.

  (* a :: p1, _  *) 
  induction p2. 
   (* a::p1, nil *)
   simpl; unfold pre_merge_add_error; rewrite pre_merge_error_eq_1; simpl; rewrite f0_R0; auto with real.
   (* a::p1, a0::p2 *)
   assert (H_p1:is_sorted p1); [apply is_sorted_cons_inv with (fst a) (snd a); rewrite <- (surjective_pairing); exact H1|].
   assert (H_p2:is_sorted p2); [apply is_sorted_cons_inv with (fst a0) (snd a0); rewrite <- (surjective_pairing); exact H2|].
   simpl in *.
   unfold pre_merge_add_error.
   rewrite pre_merge_error_eq_cons_cons.
   destruct (lt_eq_lt_dec (fst a) (fst a0)) as [[Hlt | Heq] | Hlt].
    (* 1 *)
    apply (IHp1 H_p1 (Build_Taylor_model (Build_Sparse_polynom (a0 :: p2) H2) e2)). 
    (* 2 *)
    apply Rle_trans with (inj_R (minus_up (add_up (snd a) (snd a0)) (add_down (snd a) (snd a0))) + 
                          inj_R (sum_add_up (pre_merge_error (p1,p2)))); [|apply property_2_u].
    apply Rplus_le_le_0_compat.

     generalize (snd a) (snd a0); intros x y;
     apply Rle_pos_nonneg_Rmult with (1/2); [fourier | ]; rewrite Rmult_comm;
     apply Rle_trans with (Rabs ( (inj_R  (add_near x y))- ((inj_R x)+(inj_R y)) )); [apply Rabs_pos|apply property_1]...

     apply (IHp1 H_p1 (Build_Taylor_model (Build_Sparse_polynom p2 H_p2) e2)). 
    (* 3 *)
    apply IHp2; assumption.
Qed.

Lemma sum_add_up_property: forall (sp1 sp2:Sparse_polynom F) x,  -1 <= x <= 1 ->
   Rabs (eval_Sparse_polynom sp1 x + eval_Sparse_polynom sp2 x - eval_Sparse_polynom (merge_add_near sp1 sp2) x) <=
       1 / 2 * inj_R (sum_add_up (pre_merge_add_error sp1 sp2)).
Proof.
 intros [p1 H1].
 induction p1; intros [p2 H2].
  intros  x _; simpl; unfold merge_add_near, pre_merge_add_near, pre_merge_add_error in *.
  rewrite pre_merge_eq_2, pre_merge_error_eq_2; simpl.
  stepl 0; [ rewrite f0_R0; fourier | symmetry; stepl (Rabs 0); [apply Rabs_R0|f_equal; ring]].
  
  induction p2.
   intros x _; simpl; unfold merge_add_near, pre_merge_add_near, pre_merge_add_error in *.   
   rewrite pre_merge_eq_1, pre_merge_error_eq_1; simpl.
   stepl 0; [ rewrite f0_R0; fourier | symmetry; stepl (Rabs 0); [apply Rabs_R0|f_equal; ring]].

   intros x Hx.
   unfold merge_add_near, pre_merge_add_near, pre_merge_add_error.
   assert (H_p1:is_sorted p1); [apply is_sorted_cons_inv with (fst a) (snd a); rewrite <- (surjective_pairing); exact H1|].
   assert (H_p2:is_sorted p2); [apply is_sorted_cons_inv with (fst a0) (snd a0); rewrite <- (surjective_pairing); exact H2|].
   assert (H_p1_a0_p2:=@pre_merge_sorted Datatypes.id (@add_near F) p1 (a0 :: p2)  H_p1 H2).

   unfold eval_Sparse_polynom at 3.
   unfold polynom.
   rewrite pre_merge_eq_cons_cons, pre_merge_error_eq_cons_cons.
   destruct (lt_eq_lt_dec (fst a) (fst a0)) as [[Hlt | Hlt] | Hlt].
    (* 1 *)
    rewrite (eval_Sparse_polynom_eq_1) with (H2:=H_p1).
    set (sp1:=Build_Sparse_polynom p1 H_p1).
    set (sp2:=Build_Sparse_polynom (a0 :: p2) H2).
    replace (eval_polynom ((fst a, Datatypes.id (snd a)) :: pre_merge Datatypes.id (@add_near F) (p1, a0 :: p2)) x) with 
            (inj_R (snd a) *(pow x (fst a)) + (eval_Sparse_polynom (merge_add_near sp1 sp2) x)); [|trivial].
    replace (pre_merge_error (p1, a0 :: p2)) with
            (pre_merge_add_error sp1 sp2); [|trivial].
    stepl (Rabs (eval_Sparse_polynom sp1 x +
                 eval_Sparse_polynom sp2 x -
                 eval_Sparse_polynom (merge_add_near sp1 sp2) x)) ;[|f_equal; ring].
    apply (IHp1 _ _ _ Hx).
    (* 2 *)

    simpl.
    apply Rle_trans with ( (1/2) * (inj_R(minus_up (add_up (snd a) (snd a0)) (add_down (snd a) (snd a0))) +
                                    inj_R(sum_add_up (pre_merge_add_error (Build_Sparse_polynom p1 H_p1) (Build_Sparse_polynom p2 H_p2))))).
    2: apply Rmult_le_compat_l; [ fourier | apply property_2_u].
    rewrite Rmult_plus_distr_l.
    rewrite <- Hlt.
    set (sp1:=Build_Sparse_polynom p1 H_p1).
    set (sp2:=Build_Sparse_polynom p2 H_p2).
    stepl (Rabs ( ( (inj_R (snd a) + inj_R (snd a0)) * (pow x (fst a)) -
                     inj_R (add_near (snd a) (snd a0)) * (pow x (fst a)) 
                   ) + 
                  (eval_Sparse_polynom sp1 x + eval_Sparse_polynom sp2 x - eval_Sparse_polynom (merge_add_near sp1 sp2) x) 
                ) ).
    2:f_equal; simpl; unfold merge_add_near, pre_merge_add_near; ring.
    apply Rle_trans with (Rabs ( inj_R (add_near (snd a) (snd a0)) -
                                  (inj_R (snd a) + inj_R (snd a0)) 
                               ) * Rabs (pow x (fst a)) + 
                         Rabs (eval_Sparse_polynom sp1 x + eval_Sparse_polynom sp2 x - eval_Sparse_polynom (merge_add_near sp1 sp2) x) );
    [ rewrite <- Rabs_mult;
      replace ((inj_R (add_near (snd a) (snd a0)) - (inj_R (snd a) + inj_R (snd a0))) * (pow x (fst a))) with 
      (inj_R (add_near (snd a) (snd a0))*(pow x (fst a)) - (inj_R (snd a) + inj_R (snd a0))*(pow x (fst a)));[|ring];
      rewrite Rabs_minus_sym; apply Rabs_triang; rewrite Rabs_minus_sym; apply Rabs_triang
    |].
    apply Rplus_le_compat; [| apply (IHp1 _ _ _ Hx)].
    apply Rle_trans with (Rabs(inj_R (add_near (snd a) (snd a0)) - (inj_R (snd a) + inj_R (snd a0)))).
     assert (H_xn_l:-1 <= (pow x (fst a)) );[apply pow_Rle_l_1; elim Hx; trivial|]. 
     assert (H_xn_r:(pow x (fst a))<= 1 );[apply pow_Rle_r_1; elim Hx; trivial|]. 
     assert (H_xn_abs:=Rabs_Rle_1 (pow x (fst a)) H_xn_l H_xn_r). 
     stepr (Rabs (inj_R (add_near (snd a) (snd a0)) - (inj_R (snd a) + inj_R (snd a0))) * 1); [|ring].
     apply Rmult_le_compat_l; trivial; apply Rabs_pos.

     apply property_1...

    (* 3 *)
    rewrite (eval_Sparse_polynom_eq_1) with (H2:=H_p2).

20 lines of C++

370 lines of Coq
230 spec (60%)
140 proof (40%)

Extending the TMF Calculus?

division to be added to the axiomatisation for F.

NaN and ±∞ will be added.

Operations and axioms to be updated on F + {NaN,±∞}.

anti-differentiation of Taylor models is possible using integer
division.

Further Parametrisation?

To ‘future-proof’ the framework ideally we have to develop

Polynomials: a minimal interface covering
I representation: sparse, incremental sparse, . . .
I evaluation: ordinary, Horner, . . .
I calibration: sweeping the truncation error

Function Models: a type for approx. of continuous functions.

Further Parametrisation?

To ‘future-proof’ the framework ideally we have to develop

Polynomials: a minimal interface covering

Function Models: a type for approx. of continuous functions.

The minimal interface should cover

I evaluation on floats, intervals and real numbers
I composition of approximations
I composition of a computable function and an approximation.

+ Good for transcendental functions.

